

# MCS-51 单片机的系统扩展技术（五）

## 5 数——模转换接口

在工作控制和智能化仪表中，通常由微型计算机进行实时控制及实时数据处理。计算机所加工的信息总是数字量，而被控制或测量对象的有关参量往往是连续变化的模拟量，如温度、速度、压力等等，与此对应的电信号是模拟电信号。计算机要处理这种信号，首先必须将模拟量转换成数字量，这一转换过程就是“模——数转换（A/D）”。

由计算机运算处理的结果（数字量）往往也需要转换为模拟量，以便控制对象，这一过程即为“数模转换”（D/A）。

A/D、D/A 转换技术发展极为迅速，目前常用的 A/D 或 D/A 芯片种类也非常多，本教程介绍的是比较经典的一些芯片的用法，目的在于帮助读教掌握这类芯片接口的一般方法，以及进一步理解数字系统和模拟系统的区别。当然，这些芯片本身也有一定的实用价值。

### 一、DAC 电路原理

D/A 转换是将数字量信号转换成模拟量信号的过程。D/A 转换的方法比较多，这里仅举一种权电阻 D/A 转换法的方法，说明 D/A 转换的过程。

权电阻 D/A 转换电路实质上是一只反相求和放大器，图 22 是 4 位二进制 D/A 转换的示意图。电路由权电阻、位切换开关、反馈电阻和运算放大器组成。




图 22 D/A 转换的原理

权电阻的阻值按 8 : 4 : 2 : 1 的比例配置，按照运放的“虚地”原理，当开关 D3-D0 合上时，流经各权电阻的电流分别是  $V_R/8R$ 、 $V_R/4R$ 、 $V_R/2R$  和  $V_R/R$ 。其中  $V_R$  为基准电压。而这些电流是否存在则取决于开关的闭合状态。输出电压则是：

$$V_O = - (D_3/R + D_2/2R + D_1/4R + D_0/8R) \times V_R \times R_F$$

式中 D3-D0 是输入二进制的相应位，其取值根据通断分别为 0 或 1。显然，当 D3-D0 在 0000-1111 范围内变化时，输出电压也随这发生变化，这样，数字量的变化就转化成了电压（模拟量）的变化了。这里，由于仅有 4 位开关，所以这种变化是很粗糙的，从输出电压为 0 到输出电压为最高值仅有 16 档。显然，增加开关的个数和权电阻的个数，可以将电压的变化分得更细。一般，至少要有 8 个开关才比较实用。8 个开关，就意味着输出量从最小（0）到最大一共被分成 256 档了。

上面的这种 D/A 转换技术对于权电阻的精度要求是相当高的，由于权电阻的大小并不

相同，所以制造很困难，因此，最常用的是另一种结构是被称之为 R-2R 型电路 DAC 的结构。关于这种 DAC 转换的原理，不再分析，只要知道数字量是可以转换为模拟量就行。

## 二、DAC 接口电路

D/A 转换器有各种现成的集成电路。对使用者而言，关键是选择好合用的芯片以及掌握芯片与计算机的正确连接方法。下面以常用的 DAC0832 为例作一说明。

DAC0832 是 CMOS 工艺制造的 8 位单片 D/A 转换器，其引脚图和逻辑图如图 23 所示。

DAC0832 主要由两个 8 位寄存器和一个 8 位 D/A 转换器组成。使用两个寄存器的好处是能简化某些应用中的硬件接口电路设计。




图 23 DAC0832 的引脚图及内部结构框图

该 D/A 转换器为二十引脚双列直插封装，各引脚含义如下：

DIO-DI7：数字量输入端。

ILE：数据锁存允许信号，高电平有效；

$\overline{CS}$ ：输入寄存器选择信号，低电平有效；

$\overline{WR1}$ ：输入寄存器的写选通信号 1，低电平有效。和  $\overline{CS}$ 、ILE 配合，构成第一级输入锁存；

$\overline{WR2}$ ：输入寄存器的写选通信号 2，低电平有效；

$\overline{XFER}$ ：数据转移控制信号，输入低有效，和  $\overline{WR2}$  配合构成第二级锁存；

$V_R$ ：基准电压输入线；

$R_{FB}$ ：反馈信号输入线，在芯片内部已有反馈电阻；

$I_{OUT1}$  和  $I_{OUT2}$ ：电流输出线。 $I_{OUT1}$  和  $I_{OUT2}$  的和为常数， $I_{OUT1}$  随 DAC 寄存器的内容线性变化；

VCC：工作电源；

DGND：数字地；

AGND：模拟信号地。D/A 转换芯片输入的是数字量，输出为模拟量，模拟信号容易受到干扰，所以采用高精度的  $V_R$  和独立的模拟地，以获得最好的效果。当然，这两种地最终还是要接在一起的，在布置印制板时，将这两路地线在电源处进行一点接地的处理。

DAC0832 是电流型输出，应用时要外接运算放大器使之成为电压型输出。

## 三、DAC0832 的应用

### (1) 单缓冲方式应用

图 24 是 DAC0832 的典型应用系统，系统只有一路输出，采用单缓冲方式。在这种方式下，将二级寄存器的控制信号并接即可。输入数据在控制信号的作用下，直接进入 0832 的控制寄存器。

图 24 中, ILE 接+5V, 片选信号  $\overline{CS}$  和传送信号  $\overline{XFER}$  都连到地址线 A0, 这样输入寄存器和 DAC 寄存器的将被同时选中。写选通线  $\overline{WR1}$  和  $\overline{WR2}$  都和 8031 的写信号  $\overline{WR}$  相连, CPU 对 0832 执行一次写操作, 则把一个数据直接写入 DAC 寄存器。0832 的输出模拟信号随之对应变化。

D/A 转换器的基准电压由稳压管提供, 如果要求更高, 可以使用一些高精度的稳压源(如 TL431) 来提供。

下面的一些程序将在运放输出端产生程控波形。

锯齿波发生器:

```

START: MOV  DPTR, #0EFFFH ; 0832 的端口地址
      MOV  A, #00H      ; 初值送 0 (无输出)
LOOP:  MOVX @DPTR, A   ; 将 A 中的值送到 0832 转换
      INC  A            ; A 中的值加 1
      AJMP LOOP        ; 循环
    
```




图 24 DAC0832 的应用

上面的程序中, A 的值从一开始的 0 开始加, 由 0 变到 FFH, 输出电压也随之由 0V 变到最大输出, 然后 A 中的值 (FFH) 加 1 之后变成 0, 输出电压又加到 0, 这样就形成了一个锯齿波。

### (2) 双缓冲方式应用

DAC0832 也可以方便地工作于双缓冲模式, 这适用于同时输出两路或多路模拟信号, 并且这些信号必须同步的情形, 详情请参考有关资料。

## 5.6 模——数转换接口

模——数转换电路的种类很多, 选择 A/D 转换器件主要从速度、精度和价格等方面进

行考虑。根据 A/D 转换器的工作原理，可以分为下面的三种类型：

(1) 并行 A/D 变换器：速度高，价格也很昂贵，用于高速（如视频处理）场合。

(2) 逐次逼近型 A/D 转换器：精度、速度、价格方面比较折衷，是最常用的一种 A/D 转换器件。

(3) 双积分型 A/D 转换器：精度高，抗干扰能力强，价格低，但是速度慢，常用于测量仪表等场合。

A/D 技术发展很快，各种新型的 A/D 层出不穷，A/D 的选择要考虑很多因素，和使用者的基本功、见识等有很大的关系。本教程主要是通过典型芯片的介绍，让读者了解单片机与 AD 转换芯片接口的一般的方法，目的还是在于加深对单片机系统的理解。

### 一、MCS-51 配置 ADC0809 扩展 A/D 转换器

ADC0809 是最常用的 8 位 A/D 转换器，属逐次逼近型。ADC0809 由单一 +5V 电源供电，片内有带锁存功能的 8 路模拟电子开关，可对 0-5V 8 路的输入模拟电压信号分时进行转换，完成一次转换约需 100us，输出具有 TTL 三态锁存缓冲器，可直接与 MCS-51 单片机的总线相连。

#### (1) ADC0809 内部结构与外特性

ADC0809 是 CMOS 工艺逐次逼近型 A/D 转换器，其字长为 8 位，可 8 路输入模拟量分时转换的组成芯片。图 25 是 ADC0809 的外部引脚配置图，图 26 是 ADC0809 的内部结构框图。




图 25 ADC0809 的引脚图




图 26 ADC0809 的内部结构框图

从图 5-26 可以看出，ADC0809 片内集成了 8 路模拟多路开关、地址锁存器与译码、8 位 A/D 转换器以及 8 位三态输出锁存器四部份组成。各引脚的功能含义如下：

D0~D7：8 位二进制数字量输出端口；

IN0~IN7：8 路模拟量输入端口

VCC：+5V 工作电压源。

GND：接地端；

VREF (+) 和 VREF (-)：基准参考电压，决定了输入模拟量的量程范围；

CLK：时钟信号输入端，时钟频率决定了转换速度，转换一次需要 64 个时钟周期。

START：A/D 转换启动信号输入端口，高电平有效；

ALE：地址锁存允许信号输入端口，ALE 的下降沿将地址打入锁存器；  
 EOC：A/D 转换结束信号输出端口，开始转换时为低电平，一旦转换结束输出高电平；  
 OE：完成转换后数字量输出允许控制信号输入端口，高电平有效，用以打开三态数据锁存器的输出。

A、B、C：地址输入端口。用三位编码组成 3——8 译码输出，选通 8 路模拟电子开关，实现 IN0~IN7 的 8 路选 1 工作。A、B、C 三位地址的输入与 8 路通道的对应关系如下：

地址 编码	C	0	0	0	0	1	1	1	1
	B	0	0	1	1	0	0	1	1
	A	0	1	0	1	0	1	0	1
选中通道	IN0	IN1	IN2	IN3	IN4	IN5	IN6	IN7	

## 二、MCS-51 单片机配置 ADC0809 的硬件接口

图 27 是 ADC0809 与 MCS-51 单片机的接口电路图。

地址线与数据线的连接：ADC0809 的输出部份有三态缓冲器，可以直接和数据总线相连；ADC0809 的地址选择信号线 ADDA、ADDB、ADDC 可以与地址线中的任意三根相连（图中与 AD0、AD1、AD2 相连），这样 8 个输入端所对应的地址就是从：

xxxx xxxx xxxx x000~xxxx xxxx xxxx x111

也就是 FFF0H~FFF7H

时钟线的连接：ADC0809 工作时必须外接时钟，如果单片机的主振频率为 6M，就可以直接借用 ALE 信号来作为 ADC0809 的时钟信号，当不使用 MOVX 类指令时，ALE 是时钟频率的 6 分频，在 6M 晶振的频率下，ALE 的频率是 1M，如果用了更高频率（如 12M）的晶振，用 ALE 作为 ADC0809 的时钟就不恰当了。可以把 ALE 二分频之后再提供给 ADC0809，也可以用 555 之类的振荡电路做一个时钟发生器。

控制信号的连接：ADC0809 所要求的或者所提供的都是高电平有效信号，而 MCS-51 能够提供或者需要的却是低电平有效信号，所以必须用一些门电路进行转换。具体的接法参考图 27，转换结束信号 EOC 通过非门与 MCS-51 的  $\overline{INT1}$  相连，每次转换完毕进行中断请求，MCS-51 单片机在中断程序中读取数据，也可以用软件查询的方法获得数据。

例：利用中断方式编写程序，将 8 个模拟量信号轮流采集一遍，并依次将转换后的结果放在 MCS-51 单片机的 30H 开始的单元中。

```

ORG    0000H
JMP    MAIN
ORG    13H          ; 外中断 1 入口地址
JMP    ADINT       ; 转到中断服务程序入口
ORG    30H
MAIN:  MOV    R0, #30H      ; 存储区首地址
      SETB   IT1          ; 下降沿中断
      SETB   EA           ; 开总中断
      SETB   EX1         ; 开外部中断 1
      MOV    R3, #08H     ; 共有 8 个通道
      MOV    DPTR, #0FFF0H ; 指向通道 0
      MOVX  @DPTR, A      ; 启动转换
LOOP:  SJMP  LOOP        ; 等待转换的结果
      ; 以下中断处理
ADINT: PUSH  PSW

```

```

    PUSH   ACC           ; 入堆栈保护
    MOVX   A, @DPTR     ; 读转换结果
    MOV    @R0,A        ; 存转换结果
    INC    DPTR         ; 指向下一通道
    INC    R0           ; 存储数据区指针加 1
    DEC    R3           ; 减去 1 次
    CJNE   R3, #0, AD1  ; 和 0 相比, 如果未到则转
    MOV    DPTR, #0FFF0H ; 复原, 指向通道 0
    MOV    R0, #30H     ; 指向数据区首地址
    MOV    R3, #8       ; 转换次数恢复
    JMP    ADRET        ; 不再启动转换, 直接返回
AD1:  MOVX   @DPTR, A    ; 启动转换
ADRET: POP    ACC
      POP    PSW        ; 出栈
      RETI             ; 返回

```

中断式数据采集比较节约主程序的时间, 每一个通道的数据采集完成之后, 启动下一次转换, 然后返回主程序, 不用等 8 个通道全部完成。当 8 通道的采集工作全部完成, 把所有的指针指向它们原来的位置, 不再启动下一次转换。一般主程序是一个循环, 会由主程序再次启动下一轮 8 个通道的数据采集工作。除了可以用中断的方式之外, 还可以用查询的方式进行转换。

例: 条件同上, 但用查询方式编程。下面以子程序的形式给出。

```

    MOV    DPTR, #0FFF0H ; 指向通道 0
    MOV    R0, #30H     ; 数据区首地址
    MOV    R3, #08H    ; 通道数
LOOP: MOVX   @DPTR, A    ; 启动转换
HERE:  JB    P3.3, HERE ; 查询一次转换是否结束
MOVX   A, @DPTR        ; 读转换结果
    MOV    @R0, A      ; 存转换结果
    INC    R0          ; 指向下一个存储单元
    INC    DPTR        ; 指向下一个通道
    DJNZ   R3, LOOP    ; 8 路数据未采集完, 继续
RET

```

用这种编程方法，反相器可以省略，也不一定非要接在  $\overline{INT1}$  引脚上，接在任何一个 I/O 引脚都可以。从上面的两个例子中还可以看到 P3 口的第二功能其实并没有什么特殊之处，关键看你把它当成是什么，第一个例子是当成外中断 1 来用，而第二个例子却是当成了普通的 I/O 口输入来用，如此而已。

如果对于转换的时间要求不高，还可以省掉 EOC 接口，直接用延时的方法来读取转换的结果。方法是在启动转换之后延时一段时间，然后去读 0809 的转换结果，这一段时间根据接在 0809 的 CLK 端的振荡频率及需要 64 个时钟来完成一次转换进行计算，然后适当留一些余量。这样可以省去一个反相器和少占用一个主机的 I/O 口。




图 27 ADC0809 与 MCS-51 单片机连接